
Листинг программы с комментариями. Программа может программироваться без изменений в микроконтроллеры PIC12C508, PIC12C508A, PIC12C509, PIC12C509A.


list      p=12c509a


#include <p12c509a.inc>


__CONFIG _CP_OFF & _WDT_ON & _IntRC_OSC & _MCLRE_OFF

; Переменные

ALL_VAR
UDATA

curcnt    RES 1   ; Текущее значение

curout    RES 1   ; Текущее выводимое значение

outdl     RES 2   ; Задержка вывода

secdl     RES 1   ; Задержка в секундах

keydl     RES 1   ; Пауза реагирования на клавишу

impcnt    RES 1   ; Счетчик импульсов мигания

MAIN CODE 0x00


goto start
; Переход на основную программу

; Тут - информация в открытые 32 байта PIC12C508A


dt "Cubic V1.0 by ALB alb@crowd.ru (C)CROWD Russia 2005"

#define KEY GPIO,3

PROC CODE

start


movwf OSCCAL
restart

; Конфигурация входов-выходов

  movlw b'00001000'

  tris 6

  movlw 0xff

  movwf GPIO

; Конфигурация регистра OPTION_REG

; Просыпание, подтяжка уровня, 

; таймер на внутреннем генераторе, максимальное значение WDT


movlw b'00011111'

;


option

; Смотрим, надо ли просыпаться?

  btfss KEY
  goto startcount
; Нет, просыпаться не надо - спим дальше

  clrwdt

  movlw B'00010000'

  movwf STATUS

  movf GPIO,0

  sleep
  goto restart
;

; ### Первая стадия - считаем, пока нажата клавиша
;

; Начинаем считать

startcount

  movlw .1

  movwf outdl+1

  clrf outdl
; Запускаем счет и вывод

nextcount

  clrwdt

  decfsz outdl,1

  goto aroundoutdelay2

  decfsz outdl+1,1

  goto aroundoutdelay

; Выводим текущее выводимое  

  incf curout,0       ;1

  andlw 7             ;2

  movwf curout        ;3

  call getscrstate    ;4,5,6,7,8,9

  movwf GPIO          ;10

  movlw .15           ;11

  movwf outdl+1       ;12

toinccnt

; Увеличиваем текущее состояние

  incf curcnt,1

  movlw 6

  subwf curcnt,0

  btfsc STATUS,C
  clrf curcnt

; Проверяем состояние клавиши

  btfss KEY

  goto nextcount  ; Общая длительность цикла - 25 команд (мкс)

;

; #### Вторая стадия - досчитываем до нужного состояния

;

; Клавишу отпустили - досчитываем до нужного состояния  

  clrf outdl

  movlw .100

  movwf outdl+1

wincycle

; Пауза в 0.2 секунды 

; Проверяем кнопку

  btfss KEY

  goto nextcount    ; Нажали кнопку - продолжаем считать

; Выводим картинку

  clrwdt

  decfsz outdl,1

  goto wincycle

  decfsz outdl+1,1

  goto wincycle

; Выводим текущее выводимое  

  incf curout,0       

  andlw 7             

  movwf curout        

  call getscrstate    

  movwf GPIO          

  movlw .100          

  movwf outdl+1

; Проверяем, не совпадает ли с нужным?

  movf curout,0

  call getnumber

  subwf curcnt,0

  btfss STATUS,Z
  goto wincycle

; Последняя выведенная картинка соответствует выпавшему числу

;

; ### Третья стадия - показываем картинку 10 секунд 

;                             или до нажатия кнопки

;

; Пауза в десять секунд, нажатие на кнопку прерывает паузу

  clrf keydl

  movlw .20         ; Счетчик полусекундных интервалов

  movwf secdl

secdelay

  movlw .180

  movwf outdl+1

secdelay2   ; Общая длительность цикла должна быть 13 команд

  btfsc KEY         ;1

  goto notkeypress  ;2

  incf keydl,1      ;3

  btfsc keydl,7     ;4

  goto off10second    ;5

  goto aroundkeypress ;6,7

;

notkeypress

  goto $+1

  goto $+1

  clrf keydl

aroundkeypress

  nop               ;8

  clrwdt            ;9

  decfsz outdl,1    ;10,11

  goto secdelay2    ;12,13

  decfsz outdl+1,1

  goto secdelay2

  decfsz secdl,1

  goto secdelay
off10second
;

; ##### Четвертая стадия - мигаем несколько раз последним значением

;

; Десять секунд кончились - мигаем четыре раза

  movlw .8

  movwf impcnt

; 

flashcycle

  movf curout,0

  call getscrstate

  movwf GPIO

  movlw 0xff

  btfss impcnt,0    

  movwf GPIO

; Пауза в 0.2 секунды
  movlw .80

  movwf outdl+1

flashdelay1

  clrwdt

  decfsz outdl+1,1

  goto flashdelay1

  decfsz outdl,1

  goto flashdelay1

  decfsz impcnt,1

  goto flashcycle

; Перезапускаем весь процесс
  goto restart

getscrstate

  addwf PCL,1

  retlw b'11011111'

  retlw b'11111101'

  retlw b'11011011'

  retlw b'11111001'

  retlw b'11011001'

  retlw b'11111000'

  retlw b'11111011'

  retlw b'11011101'

getnumber

  addwf PCL,1

  retlw 0

  retlw 1

  retlw 2

  retlw 3

  retlw 4

  retlw 5

  retlw 1

  retlw 3

; Компенсация второго вычитания

aroundoutdelay2

  nop

  nop  

; Компенсация вывода на светодиоды

aroundoutdelay
  goto $+1      ;2,3

  goto $+1      ;4,5

  goto $+1      ;6,7

  goto $+1      ;8,9

  nop           ;10

  goto toinccnt ;11 (12)


END

